



河南大學
Henan University

汇编语言与接口技术

—— 第 8 章 中断与中断控制器 8259A

主讲教师：舒高峰


电子邮箱：gaofeng.shu@henu.edu.cn

联系电话：13161693313

目录

- 01 中断系统概述
- 02 8086/8088 的中断系统
- 03 可编程中断控制器 8259A

01 | 中断概述 - 8086管脚



01 | 中断概述-中断的基本概念


定义

- 中断是指 CPU 正常运行程序时，由**系统内/外部非预期事件**或**程序中预先安排好的指令性事件**引起的，CPU **暂停**当前程序的执行，**转去**为该事件服务的程序中执行，服务完毕后，再**返回**原程序继续执行的过程

注意

- **中断**是一个 CPU 执行程序的变化过程
- 所有能引起中断的事件均称为**中断源**
- 处理中断事件的**中断服务程序**是预先设置的
- 结束中断返回原程序时，要以**原状态返回**暂停处继续执行

01 | 中断概述-中断过程示意



01 | 中断概述 - 中断的处理过程

中断请求

- CPU 获得软硬件的中断请求

中断响应

- CPU 获得中断类型号 n ，转入中断服务程序的过程

中断服务

- 执行中断服务程序

中断返回

- CPU 执行 **IRET 指令** 时，自动产生中断返回

由硬件实现

01 | 中断概述 - 中断的处理过程 - 中断请求

中断请求

- CPU 在每个指令周期的最后一个时钟周期，采样各中断请求
 - 优先顺序为
内部中断 (除单步) → NMI 中断 → INTR 中断 → 单步中断

中断响应

- 若有中断请求，且允许响应，则直接进入中断响应处理，否则 (无请求或不允许)，继续按顺序执行下一条指令
 - 若是 INTR 中断，且 $IF=1$ ，则 CPU 响应该 INTR 中断
 - 若是单步中断，且 $TF=1$ ，则 CPU 响应该单步中断
 - 若是 NMI 中断或内部中断，则 CPU 立即响应该中断

01 | 中断概述 - 中断的处理过程 - 中断服务

关中断

- 防止中断响应过程被其他中断请求打断，自动清零 IF

获取中断类型号 n

- 从中断向量表中取出对应**中断向量** (中断服务程序入口地址)
 - 若为**内部中断**或 NMI 中断，则 CPU 直接取其**固定的类型号**
 - 若为**指令中断**，则 CPU 直接取**操作数字段**作为其类型号
 - 若为 INTR 中断，则 CPU 进入**中断响应周期**，**从中断源获取**中断类型号


注意

- 入栈**原 CS、IP 的值及断点状态信息**，**保护现场**
- 将中断向量装入 CS、IP 寄存器，**转入中断服务程序**执行

01 | 中断概述 - 中断的处理过程 - 中断响应周期

中断响应周期

- CPU 通过**两个连续的总线周期**向中断源发**两次** INTA# 应答信号
 - INTA#₁: **通知被响应的外设**, 准备提供中断类型号
 - INTA#₂: CPU 从系统数据线上**读入**中断源准备好的**中断类型号**



01 | 中断概述 - 中断服务程序的注意事项

中断嵌套的控制

- 若中断期间允许中断嵌套，则中断服务程序中应先**开中断**
 - 使用 **STI 指令**，使 $IF=1$

注意维护主程序的数据一致性

- 一般在中断服务程序中应**保护所用到的寄存器**
- 同子程序中的寄存器**保护与恢复**

中断服务程序返回

- 中断服务程序**最后执行的一条指令**必须是 **IRET**
 - 其物理位置不一定是最后一条指令

01 | 中断概述-中断优先权

中断优先权

- 系统按照**中断源的性质**以及**中断处理的紧急程度**，为每个中断源指定优先权，用于决定 CPU 对中断源的响应顺序
- 当有多个中断源同时发出请求时，**中断源的优先权**决定了**中断请求是否能够被响应**

优先权管理方式

- 固定优先权
 - 各中断源的优先权固定，始终保持不变
- 循环优先权 (实际上是一种**等优先权方式**)
 - 每结束一次中断服务时，自动循环各中断源的优先权

具体的优先级判断方式在后续课程中介绍

02 | 中断系统-中断指令 INT n

软中断指令 INT (Interrupt) 指令格式

- INT n
 - n: 中断类型码, 0~255

功能

- 执行 n 号中断服务程序, 其入口地址由 [0000:4n] 开始的 4 个存储单元指定

注意

- 中断指令执行时会自动完成以下工作:
 - 标志寄存器和断点地址压入堆栈保存
 - 关中断, 即令 $IF=0$ 、 $TF=0$
 - 取中断服务程序的起始地址 (IP)=[0000:4n],
CS=[0000:4n+2]

02 | 中断系统-溢出中断指令 INTO

溢出中断指令 INTO (Interrupt on Overflow) 指令格式

- INTO

功能

- 检测 OF 位。若 OF=1, 则等价于 INT 4, 即执行 4 号中断服务程序; 若 OF=0, 则无操作

注意

- 溢出中断指令执行时会自动完成以下工作:
 - 标志寄存器和断点地址压入堆栈保存
 - 关中断, 即令 IF=0、TF=0
 - 取中断服务程序的起始地址 (IP)=[0000H:0010H], CS=[0000H:0012H]

02 | 中断系统-中断返回指令 IRET

中断返回指令 IRET (Interrupt Return) 指令格式

- IRET

功能

- 结束中断服务程序，返回主程序

注意

- 中断返回指令执行时会自动完成以下工作：
 - 断点地址弹出堆栈
 - 标志寄存器弹出堆栈

目录

- 01 中断系统概述
- 02 **8086/8088 的中断系统**
- 03 可编程中断控制器 8259A

02 | 中断系统-中断向量和中断类型号

中断向量

- 中断服务程序的**入口地址**
- 共 4 个字节 —— 中断服务程序的**段地址**和段内**偏移地址**

中断向量表

- 集中存放系统中所有中断向量的存储区
- 8086 PC 机中，将存储器物理地址为 0~3FFH 的 1024 个单元作为中断向量表，可容纳 256 个中断向量

中断类型号

- 每个中断向量在向量表中的位置编号，0~255

02 | 中断系统-常用的中断类型号

常用中断类型号及其功能


| 中断类型号 | 中断功能 |
|-------|-------------|
| 0H | 除法错中断 |
| 1H | 单步中断 |
| 2H | NMI |
| 3H | 断点中断 |
| 4H | 溢出中断 |
| 5H | 打印屏幕 |
| | |
| 20H | 结束程序并返回 DOS |
| 21H | 请求 DOS 功能调用 |
| 22H | 结束地址 |
| | |

02 | 中断系统-中断向量和中断类型号

中断向量和中断类型号的关系

- 中断类型号 $(n) \times 4 =$ 中断向量在表中的偏移地址
 - 如: $n = 8$, 则应从向量表 $20H \sim 23H$ 中取出中断向量

CPU 使用向量中断的过程



02 | 中断系统-中断源类型

中断源类型



02 | 中断系统 - 中断源类型 - 不可屏蔽中断 NMI

不可屏蔽中断 NMI

Non Maskable Interrupt

- 由 CPU 的 **NMI 引脚**引入该中断请求，CPU 得到该中断请求**立即响应**，不需要 CPU 发响应信号
- 中断类型号固定为 **2**
- 中断优先级高于可屏蔽中断 INTR，一般用于**系统对紧急情况的处理，用户不能使用**
- 常见的 NMI 中断：
 - RAM 奇偶校验错、I/O 通道校验错、协处理器出错等

02 | 中断系统 - 中断源类型 - 可屏蔽中断 INTR

可屏蔽中断 INTR

Interrupt Request

- 由 CPU 的 **INTR 引脚**引入该中断请求，该中断请求只有在 CPU 的中断允许标志位 **IF=1** 时，CPU 才会通过发回**响应信号 (INTA)** 的方式响应
- CPU 对 INTR 中断的响应控制可由 **STI (IF=1)**、**CLI (IF=0)** 两条指令完成
- 此类中断的中断类型号由**中断源**提供
- 常见的 INTR 中断：
 - 外部设备的中断请求，如 I/O 设备的数据传送请求，用户可自行设计

02 | 中断系统 - 中断源类型 - 指令中断 INT n

指令中断 INT n

- 由 CPU 执行在程序中**预先安排好的**中断指令引起的
 - 该中断是**可预期的**的、且**不可屏蔽**的
- 处理该中断时，**CPU 不需要发响应信号**
- 指令的操作数字段 (n) 即为中断类型号
 - **中断类型号的范围为 $5 \leq n \leq 255$**
- 这类中断包括 **BIOS 中断**、**DOS 中断**以及一些未定义的自由中断 (可由系统扩充或根据应用需要自定义)

02 | 中断系统 - 中断源类型 - 软件故障中断

软件故障中断

- 在程序运行过程中，出现**错误**而引起的中断
 - **不可预期的中断**
- **除法错中断**
 - **中断类型为 0**
 - 中断的原因
 - ◆ 执行除法指令时除数为 0，或除得的商大于规定位数
- **溢出中断**
 - **中断类型为 4**
 - CPU 运算时，若结果超出数据表示范围，则置 OF=1，引发溢出中断，跳出当前程序的执行

02 | 中断系统-中断源类型-调试中断

调试中断

- 程序调试过程中，人为设置的中断
 - 可预期的中断
- 单步中断
 - 中断类型号为 1
 - 单步调试程序时，若 $TF=1$ ，则在每条指令后自动加一条 1 号中断指令，产生单步中断
- 断点中断
 - 中断类型号为 3
 - 断点调试程序时，则指定指令后自动加一条 3 号中断指令

02 | 中断系统-中断向量表的读写

中断向量表的读写

- 系统的**中断向量**是在开机上电时，由**系统软件**装入内存中的中断向量表中的
- 用户使用自定义的中断，则必须
 - **编制中断服务程序**，保存于存储器中
 - 将该中断的**中断向量装入中断向量表**中
 - 根据其**中断类型号**设置相应的硬件连接
 - 使用软件或硬件请求的方式**调用中断服务程序**

02 | 中断系统-中断向量读写的DOS功能调用

写入中断向量

- 功能：将 (DS:DX) 中内容写入向量表中的指定位置
- 入口参数
 - AH — 功能号 (25H)、AL — 中断类型号、DS:DX — 中断向量
- 出口参数：无

读出中断向量

- 功能：将向量表中指定位置的向量读出至 (ES:BX) 中
- 入口参数
 - AH — 功能号 (35H)、AL — 中断类型号
- 出口参数
 - ES:BX — 指定中断类型号的中断向量值

02 | 中断系统-中断向量的修改

中断向量表的修改

- 为避免系统资源冲突，一般采用中断向量修改的方式来使用中断向量
- 中断向量修改的方法：利用 DOS 功能调用
 - 利用 **35H 号**功能调用，**获取原中断向量**，并保存
 - 利用 **25H 号**功能调用，**将新中断向量写入向量表**
 - 新中断服务程序执行完毕后，再使用 **25H 号**功能调用**恢复原中断向量**

02 | 中断系统-中断向量的修改举例

示例：采用中断向量修改的方式为中断服务程序 INTR1 设置中断类型为 60H 的中断向量

● 程序段如下：


```
1. ; 获取旧中断向量
2. MOV AH, 35H
3. MOV AL, 60H
4. INT 21H
5. MOV OLD_OFF, BX
6. MOV OLD_SEG, ES
7. ....
```

BX IP_{60H}

ES CS_{60H}

中断向量表

程序数据段




02 | 中断系统-中断向量的修改举例

示例：采用中断向量修改的方式为中断服务程序 INTR1 设置中断类型为 60H 的中断向量

● 程序段如下：

```
1. ;写入新中断向量
2. MOV AH, 25H
3. MOV AL, 60H
4. MOV DX, SEG INTR1
5. MOV DS, DX
6. MOV DX, OFFSET INTR1
7. INT 21H
8. ;..... ;中断服务程序
9. ;恢复原中断向量
10. MOV AH, 25H
11. MOV AL, 60H
12. MOV DX, OLD_SEG
13. MOV DS, DX
14. MOV DX, OLD_OFF
15. INT 21H
```



目录

- 01 中断系统概述
- 02 8086/8088 的中断系统
- 03 可编程中断控制器 8259A

03 | 中断控制器-子目录

8259A 中断控制器

- 8259A 的工作特点
- 8259A 的引脚和内部结构
- 8259A 的工作方式
- 8259A 的编程
 - 8259A 的命令字和初始化编程
- 8259A 的应用
 - 8259A 的单片和级联使用
- 8259A 的应用举例



03 | 中断控制器-8259A的工作特点

工作特点

- **接收和扩充外设的中断请求**
 - 单片工作时，每片 8259A 可接收 8 个外设中断
 - **两级级联**时，最多可用 9 片 8259A 管理 64 个外设中断
- **对外设中断优先权的进行排队管理**
 - 8259A 可采用**固定**、**循环**等多种优先权管理方式
- **向 CPU 提供中断类型号**
 - 通过**软硬件配合**的方式，由 8259A 向 CPU 提供**中断类型号**
- **屏蔽和开放中断请求**
 - 不同的中断屏蔽方式，实现对**特定**或**全部**中断源的屏蔽


03 | 中断控制器-子目录

8259A 中断控制器

- 8259A 的工作特点
- 8259A 的引脚和内部结构
- 8259A 的工作方式
- 8259A 的编程
 - 8259A 的命令字和初始化编程
- 8259A 的应用
 - 8259A 的单片和级联使用
- 8259A 的应用举例

03 | 中断控制器-8259A的外部引脚

8259A的外部引脚



03 | 中断控制器 - 8259A 的外部引脚1

面向 CPU 的引脚

Interrupt Acknowledge

- 双向数据总线 $D_0 \sim D_7$
- 输入读写控制信号 $WR\#$ 、 $RD\#$
- 输入片选信号 $CS\#$
- 输入地址总线 A_0
- 中断请求和响应信号 INT 、 $INTA\#$
 - 8259A 与 CPU 的**中断请求**和**应答联络**信号

03 | 中断控制器 - 8259A的外部引脚2

其他引脚


Interrupt Request
Slave Program
Enable Buffer
Cascade

- 中断请求信号 $IR_0 \sim IR_7$
 - 接外设中断源的中断请求信号
- 双功能级联引脚 $SP\#/EN\#$
 - 非缓冲连接方式，为 $SP\#$ 信号 (级联输入信号)
 - ◆ 用于区分 8259A 的主从片 (0 - 从片, 1 - 主片)
 - 缓冲连接方式，为 $EN\#$ 信号 (数据收发允许输出信号)
 - ◆ 用于控制总线缓冲器的数据传送
- 级联总线 $CAS_0 \sim CAS_2$
 - 8259A 级联时，主从片的连接总线，主片输出，从片输入

03 | 中断控制器-系统总线连接方式

非缓冲方式


- 8259A **直接**与系统数据总线相连
- 适用于 8259A 数目较少的系统



03 | 中断控制器-系统总线连接方式


缓冲方式

- 8259A 通过**总线缓冲器**与系统数据总线相连
- 适用于多片 8259A 级联的系统



03 | 中断控制器-8259A的内部结构

8259A的内部结构



03 | 中断控制器-三个重要的寄存器1

3 个 8 位寄存器

- 3 个 8 位寄存器，每一位对应着一个中断源

| IR ₇ | IR ₆ | IR ₅ | IR ₄ | IR ₃ | IR ₂ | IR ₁ | IR ₀ |
|-----------------|-----------------|-----------------|-----------------|-----------------|-----------------|-----------------|-----------------|
| D ₇ | D ₆ | D ₅ | D ₄ | D ₃ | D ₂ | D ₁ | D ₀ |

中断请求寄存器

- 记录各中断源的中断请求状态 $\left\{ \begin{array}{l} 0 \text{ 没有请求, 或已经响应} \\ 1 \text{ 有请求且还未被响应} \end{array} \right.$

中断服务寄存器

- 记录当前 CPU 为各中断源服务的情况 $\left\{ \begin{array}{l} 0 \text{ 没有被服务} \\ 1 \text{ 正在被服务} \end{array} \right.$

中断屏蔽寄存器

- 记录 8259A 对各中断源的屏蔽信息 $\left\{ \begin{array}{l} 0 \text{ 允许中断} \\ 1 \text{ 禁止中断} \end{array} \right.$

03 | 中断控制器-三个重要的寄存器2

中断请求寄存器 IRR

Interrupt Request Register

- 其置复位操作由**硬件**自动完成
 - IR_i 的有效信号置位, $INTA\#$ 的第一个负脉冲复位
- 根据硬件信号的情况, 可多位同时有效

中断服务寄存器 ISR

In-Service Register

- 置位操作由**硬件**完成, 复位操作大多由**软件**完成
 - $INTA\#$ 的第一个负脉冲置位, 程序中的 **EOI 命令**复位
- 允许中断嵌套时, 该寄存器可多位同时被置 1

中断屏蔽寄存器 IMR

Interrupt Mask Register

- 其置复位操作由**软件**完成

03 | 中断控制器-优先级判别器PR


优先级判别器 PR

Priority Resolver

- PR 根据当前中断源优先级顺序，检查有请求中断源的**优先级**，并和**当前正在执行的中断的优先级**比较，以此确定是否向 CPU 提出新的中断请求
 - 需要配合使用 IRR、ISR 和 IMR
- PR 的结构相当于一个**优先级编码器**和一个**比较器**电路，可实现中断判优及屏蔽的功能

03 | 中断控制器-优先级判别器PR

优先级判别器 PR



03 | 中断控制器-读写控制逻辑

读写控制逻辑

Initialization Command Words
Operation Command Words

- 对 8259A **写**初始化命令字 (ICW)、**操作命令字** (OCW)
- 对 8259A **读**取其内部状态信息

| \overline{CS} | \overline{RD} | \overline{WR} | A_0 | 功能 |
|-----------------|-----------------|-----------------|-------|--|
| 0 | 1 | 0 | 0 | 写 ICW ₁ 、OCW ₂ 、OCW ₃ |
| 0 | 1 | 0 | 1 | 写 ICW ₂ 、ICW ₃ 、ICW ₄ 、OCW ₁ |
| 0 | 0 | 1 | 0 | 读 IRR、ISR、查询字 |
| 0 | 0 | 1 | 1 | 读 IMR |

03 | 中断控制器-级联缓冲/比较器

级联缓冲/比较器

- 在级联、缓冲方式下有效

缓冲方式

- 当 8259A 和 CPU 需要传送数据时，产生有效的 **EN#** 信号，送往总线缓冲器

级联方式

- 用于保存和比较**从 8259A 的标志码**
- 若 8259A **级联使用**，当 CPU 响应从片中断时，主片通过 $CAS_0 \sim CAS_2$ 输出响应中断所属**从 8259A 的标志码**，从片输入并与自己的标志码比较，相同则工作


03 | 中断控制器-子目录

8259A 中断控制器

- 8259A 的工作特点
- 8259A 的引脚和内部结构
- 8259A 的工作方式
- 8259A 的编程
 - 8259A 的命令字和初始化编程
- 8259A 的应用
 - 8259A 的单片和级联使用
- 8259A 的应用举例

03

中断控制器-8259A的工作方式



03 | 中断控制器-设置优先级方式

固定优先权 $ICW_4 (D_4=0/1)$

- 一般全嵌套方式：只允许高级中断嵌套
- 特殊全嵌套方式：允许同级中断嵌套
 - 一般用于多片级联结构中的主 8259A

循环优先权 $OCW_2 (D_7D_6=10/11)$

- 优先权自动循环
 - 一次中断服务结束，当前结束中断源的优先权自动降为最低
- 优先权特殊循环
 - 一次中断服务结束，指定中断源的优先权自动降为最低

假设 IR_i 被响应，则各中断源的优先权为：

$IR_{i+1} \rightarrow IR_{i+2} \rightarrow \dots \rightarrow IR_7 \rightarrow IR_0 \rightarrow \dots \rightarrow IR_i$

假设指定中断源 IR_k ，则各中断源优先权为：

$IR_{k+1} \rightarrow IR_{k+2} \rightarrow \dots \rightarrow IR_7 \rightarrow IR_0 \rightarrow \dots \rightarrow IR_k$

03 | 中断控制器-结束中断处理方式

普通 EOI 命令 $ICW_4 (D_1=0)$ 、 $OCW_2 (D_5=1)$

- 复位当前 ISR 中**所有置位位中优先级最高的**对应位
- 适用于 8259A 单片工作且**优先级一般全嵌套**的场合

特殊 EOI 命令 $ICW_4 (D_1=0)$ 、 $OCW_2 (D_5=0)$

- 该 EOI 命令中**指定**要复位的 ISR 的对应位
- 适用于 8259A 级联工作，且**循环优先权**的系统中

自动 EOI 命令 $ICW_4 (D_1=1)$

- **不需要在中断服务程序中发 EOI 命令**，对 ISR 的复位操作在中断响应期间 (第二个 INTA#) **自动完成**
- 适用于 8259A 单片工作，且**不会发生中断嵌套**的场合

End of Interrupt

中断服务程序中 IRET 之前的 EOI 命令；
作用：复位 8259A 的 ISR 寄存器中的对应位

03 | 中断控制器-屏蔽中断源方式

中断源的屏蔽方式

- CPU 对于 8259A 提出的中断请求，都可以加以屏蔽控制

普通屏蔽方式 $OCW_3 (D_6=0)$

- 通过编程使 **IMR** 任意一位或几位置 0 或置 1，从而允许或禁止相应中断

特殊屏蔽方式 $OCW_3 (D_6=1)$

- 用 OCW_1 对 **IMR** 中的某一位复位时，同时也会使 **ISR** 中的相应位复位
 - 特殊屏蔽是在**中断处理程序中**使用的，用了这种方式之后，尽管系统正在处理高级中断，但对外界来讲，只有同级中断被屏蔽，而**允许其它任何级别**的中断请求

03 | 中断控制器-中断触发方式

中断触发的方式

- 中断请求信号 IR_i 的有效形式，可以是**上升沿**或**高电平**有效触发

边沿触发方式 $ICW_1 (D_3=0)$

- 将**上升沿**作为中断触发信号
 - 触发后，电平即使一直维持高，也不会引起再次中断请求

电平触发方式 $ICW_1 (D_3=1)$

- 将**高电平**作为中断请求信号
 - 中断响应之后，高电平**必须及时撤除**，否则，在 CPU 响应中断，开中断之后，会引起第二次不应该有的中断

03 | 中断控制器-子目录

8259A 中断控制器

- 8259A 的工作特点
- 8259A 的引脚和内部结构
- 8259A 的工作方式
- 8259A 的编程
 - 8259A 的 4 个初始化命令字
 - 8259A 的初始化编程
 - 8259A 的 3 个操作命令字
 - 8259A 命令字的使用
- 8259A 的应用
 - 8259A 的单片和级联使用
- 8259A 的应用举例

03 | 中断控制器-8259A的4个初始化命令字

ICW₁ —— 芯片控制的初始化命令字

- 初始化必选命令字，写入偶地址端口

ICW₂ —— 设置中断类型号的初始化命令字

- 初始化必选命令字，写入奇地址端口


ICW₃ —— 级联主从片的初始化命令字

- 级联结构时的命令字，写入奇地址端口

ICW₄ —— 方式控制字的初始化命令字

- 特殊设置时的命令字，写入奇地址端口

03 | 中断控制器-初始化命令字ICW₁



- 写入 ICW₁ 芯片即恢复初始化状态
 - 自动清除 IRR、ISR、IMR
 - 一般全嵌套方式的优先级管理方式，并使 IR₀ 优先级最高
 - 普通屏蔽方式

03 | 中断控制器-初始化命令字ICW₁举例

示例：若 8259A 采用电平触发，单片使用，需要 ICW₄，则命令字 ICW₁ 应如何设置？

- ICW₁ 命令字


| | | | | | | | |
|---|---|---|---|------|---|------|-----------------|
| 0 | 0 | 0 | 1 | LTIM | 0 | SNGL | IC ₄ |
| 0 | 0 | 0 | 1 | 1 | 0 | 1 | 1 |

- ICW₁ 命令字写入 8259A 的偶地址端口
- 设端口地址为 20H~21H:

```
MOV AL, 1BH  
OUT 20H, AL
```

03 | 中断控制器-初始化命令字ICW₂

ICW₂



- 一片 8259A 所管理的 8 个中断源的中断类型号必须是连续的
 - 高 5 位由 ICW₂ 提供, 每个芯片一个编号
 - 低 3 位由中断源的引入端 (IR_i) 决定
- 用法
 - 根据 8259A 所管理中断源的中断类型号, 初始化 ICW₂

03 | 中断控制器-初始化命令字ICW₂举例

示例 1: 初始化设置某 8259A 的 ICW₂ 为 0011 0000B, 则该 8259A 引入的 8 个中断源 IR₀~IR₇ 的中断类型号应分别是多少?

✎ 答: IR₀~IR₇的中断类型号是连续的, 应为 30H~37H


示例 2: 已知某中断源的中断类型号为 5AH, 则该中断源应接 8259A 哪个引脚, 且其 ICW₂ 又应如何设置?

✎ 答: 中断类型号 5AH (0101 1010)

- ✓ 高 5 位 01011, 决定 ICW₂ 应设置为 0101 1000B
- ✓ 低 3 位决定该中断源应接入 IR₂ 引脚

03 | 中断控制器-初始化命令字ICW₃

主片 ICW₃ 的格式



- 每一位分别对应着一个 IR 引脚
 - 1 - 表示该引脚上接有从片
 - 0 - 表示该引脚没有接从片，可以引入其他中断源

从片 ICW₃ 的格式



- 从片 ICW₃ **仅低 3 位有效**，表示该从片是由主片的哪个 IR 引脚引入 8259A 系统的，这个编码也就是从片的**标志码**

03 | 中断控制器-初始化命令字ICW₃举例

示例：8259A 的级联结构中，两个从片分别接主片的 IR₃ 和 IR₆ 引脚，则主从片的 ICW₃ 应如何设置？

- 主从片 ICW₃ 的设置不同，如下

- 主片的 ICW₃
- 从片 1 的 ICW₃
- 从片 2 的 ICW₃

| | | | | | | | |
|---|---|---|---|---|---|---|---|
| 0 | 1 | 0 | 0 | 1 | 0 | 0 | 0 |
| * | * | * | * | * | 0 | 1 | 1 |
| * | * | * | * | * | 1 | 1 | 0 |

- 相关程序段：


```
;主8259A: 20H~21H  
MOV AL, 48H  
OUT 21H, AL
```

```
;从8259A: 0A0H~0A1H  
MOV AL, 03H  
OUT 0A1H, AL
```

```
;从8259A: 0A2H~0A3H  
MOV AL, 06H  
OUT 0A3H, AL
```

03 | 中断控制器-初始化命令字ICW₄

ICW₄ Special Fully Buffered 8086/8088
 Nested Mode Mode Auto EOI Mode



无效

全嵌套方式

| | |
|---|-------|
| 0 | 一般全嵌套 |
| 1 | 特殊全嵌套 |

系统连接方式

| | |
|---|-------|
| 0 | 非缓冲方式 |
| 1 | 缓冲方式 |

缓冲方式下区分主从片

| | |
|---|----|
| 0 | 从片 |
| 1 | 主片 |

使用系统

| | |
|---|--------|
| 0 | 8 位系统 |
| 1 | 16 位系统 |

中断结束方式

| | |
|---|---------|
| 0 | 非自动 EOI |
| 1 | 自动 EOI |

03 | 中断控制器 - 8259A系统中的主从片的区分

非缓冲方式

- 8259A 的 **SP#** 引脚区分主从片
 - 输入**高电平**，表示**单片**或主片
 - 输入**低电平**，表示**从片**

缓冲方式

- 8259A 的 **SP#/EN#** 引脚作为 **EN#** 信号使用
- 由初始化命令字 **ICW₄** 的 **M/S#** 位主从片区分
 - **1** — 主片
 - **0** — 从片

03 | 中断控制器-初始化命令字ICW₄举例

示例：8086 系统中，使用两片 8259A，采用缓冲方式连接，非自动 EOI 方式，则主从 8259A 的 ICW₄ 命令字如何设置？

● 主片 ICW₄ 命令字

| 0 | 0 | 0 | SFNM | BUF | M/S# | AEOI | μPM |
|---|---|---|------|-----|------|------|-----|
| 0 | 0 | 0 | 1 | 1 | 1 | 0 | 1 |

● 从片 ICW₄ 命令字

| 0 | 0 | 0 | SFNM | BUF | M/S# | AEOI | μPM |
|---|---|---|------|-----|------|------|-----|
| 0 | 0 | 0 | 0 | 1 | 0 | 0 | 1 |

● 相关程序段：

```
;主8259A: 20H~21H  
MOV AL, 1DH  
OUT 21H, AL
```

```
;从8259A: 0A0H~0A1H  
MOV AL, 09H  
OUT 0A1H, AL
```

03 | 中断控制器-子目录

8259A 中断控制器

- 8259A 的工作特点
- 8259A 的引脚和内部结构
- 8259A 的工作方式
- 8259A 的编程
 - 8259A 的 4 个初始化命令字
 - 8259A 的初始化编程
 - 8259A 的 3 个操作命令字
 - 8259A 命令字的使用
- 8259A 的应用
 - 8259A 的单片和级联使用
- 8259A 的应用举例


03 | 中断控制器-8259A的初始化编程

初始化命令字 ICW

- 在中断系统运行之前，必须对所有的 8259A 芯片进行初始化编程，由 CPU 向 8259A 写入 2~4 个**初始化命令字 ICW**
 - ICW_1 和 ICW_2 必须设置
 - ICW_1 的 **SNGL** 位决定是否需要 ICW_3 (0-需要, 1-不需要)
 - ICW_1 的 **IC4** 位决定是否需要 ICW_4 (0-不需要, 1-需要)
 - ◆ 若按照条件写出的 $ICW_4=0$, 则不需要, 否则需要
- 初始化命令字一定是在**主程序**中设置的

03 | 中断控制器-初始化命令字设置流程图

初始化命令字设置流程图



03 | 中断控制器-8259A的初始化编程举例

示例：某 8086/8088 系统中，使用单片 8259A 作为中断控制器，工作在全嵌套方式、非自动 EOI 方式，非缓冲方式，边沿触发请求中断，IR₂ 引脚上中断源的中断类型为 0AH，试写出其初始化程序段

- 初始化所需要的命令字

| | | | | | | | | | |
|--------------------|---|---|---|---|---|---|---|---|-----|
| ■ ICW ₁ | 0 | 0 | 0 | 1 | 0 | 0 | 1 | 1 | 13H |
| ■ ICW ₂ | 0 | 0 | 0 | 0 | 1 | 0 | 0 | 0 | 08H |
| ■ ICW ₄ | 0 | 0 | 0 | 0 | 0 | 0 | 0 | 1 | 01H |

- 初始化程序：将各命令字写入对应的命令端口

- 假定该 8259A 芯片的端口地址为 20H、21H

```
;写ICW1  
MOV AL, 13H  
OUT 20H, AL
```

```
;写ICW2  
MOV AL, 08H  
OUT 21H, AL
```

```
;写ICW4  
MOV AL, 01H  
OUT 21H, AL
```

03 | 中断控制器-子目录

8259A 中断控制器

- 8259A 的工作特点
- 8259A 的引脚和内部结构
- 8259A 的工作方式
- 8259A 的编程
 - 8259A 的 4 个初始化命令字
 - 8259A 的初始化编程
 - 8259A 的 3 个操作命令字
 - 8259A 命令字的使用
- 8259A 的应用
 - 8259A 的单片和级联使用
- 8259A 的应用举例

03 | 中断控制器 - 8259A的3个操作命令字

操作控制字

- 操作控制字是在 8259A 工作过程中，根据需要写入的
- OCW_1 —— **中断屏蔽**操作命令字
 - 写入**奇地址**端口
- OCW_2 —— 设置**优先级循环**及**中断结束方式**的操作命令字
 - 写入**偶地址**端口
- OCW_3 —— **特殊屏蔽**、**8259A 读操作**的命令字
 - 写入**偶地址**端口

03 | 中断控制器 - 中断屏蔽操作命令字 OCW₁

OCW₁ 的作用

Mask

| | | | | | | | |
|----------------|----------------|----------------|----------------|----------------|----------------|----------------|----------------|
| M ₇ | M ₆ | M ₅ | M ₄ | M ₃ | M ₂ | M ₁ | M ₀ |
|----------------|----------------|----------------|----------------|----------------|----------------|----------------|----------------|

- 写 8259A 的中断屏蔽寄存器 **IMR**
 - **1** — **禁止**相应 IR_i 中断
 - **0** — **允许**相应 IR_i 中断
- 在中断服务程序中，使用 OCW₁ 可动态地改变各中断源的优先权顺序

注意

- 一般，若某系统没有完全使用所有的 IR 引脚，在初始化程序中设置了 ICW 之后，可继续写入该命令字，以控制该 8259A 系统的可用中断源

03 | 中断控制器 - 中断屏蔽操作命令字OCW₁举例

示例：设置 8259A 的 IMR，使 IR₃ 允许中断，其他引脚禁止中断

- OCW₁ 命令字

| | | | | | | | |
|---|---|---|---|---|---|---|---|
| 1 | 1 | 1 | 1 | 0 | 1 | 1 | 1 |
|---|---|---|---|---|---|---|---|

- 设端口地址为 20H~21H

```
MOV AL, 0F7H  
OUT 21H, AL
```

- 若要不影响其他中断源，仅允许 IR₃ 中断？
 - 采用“读—修改—写”的方式设置

```
IN AL, 21H  
AND AL, 0F7H  
OUT 21H, AL
```


03 | 中断控制器-OCW₂

设置优先级循环及中断结束方式的操作命令字-OCW₂ 的作用

- 非自动 EOI 方式时, 作为中断结束的 **EOI 命令**
- **改变中断优先权**的管理方式

命令字格式

Rotate
Specific EOI IR Level



03 | 中断控制器-OCW₂的命令类型

OCW₂ 的命令类型

| R | SL | EOI | 功能 |
|---|----|-----|---|
| 0 | 0 | 1 | 普通 EOI 命令, 自动复位 ISR 中优先级最高位 |
| 1 | 0 | 1 | 普通 EOI 命令, 且优先级自动循环 |
| 0 | 1 | 1 | 特殊 EOI 命令, 由 L ₂ L ₁ L ₀ 指定 ISR 的复位位 |
| 1 | 1 | 1 | 特殊 EOI 命令, 且优先级特殊循环, 由 L ₂ L ₁ L ₀ 指定 ISR 的复位位 |
| 0 | 0 | 0 | 取消优先级自动循环, 变为固定优先级 |
| 1 | 0 | 0 | 设置优先级自动循环 |
| 0 | 1 | 0 | 无效 |
| 1 | 1 | 0 | 设置优先级特殊循环, 由 L ₂ L ₁ L ₀ 指定中断源 |

03 | 中断控制器-OCW₂的命令字举例

示例 1: 若 8259A 采用普通 EOI 方式, 则如何设置其 IR₃ 中断服务程序中的 EOI 命令?

- EOI 命令用于**中断服务程序**的 IRET 之前, 以复位 8259A 内部的 ISR 寄存器
- **普通 EOI 命令**为:

| R | SL | EOI | 0 | 0 | L ₂ | L ₁ | L ₀ |
|---|----|-----|---|---|----------------|----------------|----------------|
| 0 | 0 | 1 | 0 | 0 | 0 | 0 | 0 |

- 设端口地址为 20H~21H

```
MOV AL, 20H  
OUT 20H, AL
```

- 若为**特殊 EOI 方式**, 则命令字为 **011 00 011**

03 | 中断控制器-OCW₂的命令字举例

示例 2: 若 8259A 采用普通 EOI 方式优先级自动循环方式管理中断源, 则如何设置其 IR₃ 中断服务程序中的 EOI 命令?

- 普通 EOI+优先级自动循环命令为:

| R | SL | EOI | 0 | 0 | L ₂ | L ₁ | L ₀ |
|---|----|-----|---|---|----------------|----------------|----------------|
| 1 | 0 | 1 | 0 | 0 | 0 | 0 | 0 |

- 设端口地址为 20H~21H


```
MOV AL, 0A0H  
OUT 20H, AL
```

- 若为特殊 EOI 方式, 则命令字为 111 00 011

```
MOV AL, 0E3H  
OUT 20H, AL
```

03 | 中断控制器-OCW₃

特殊屏蔽、8259A 读操作的命令字-OCW₃



03 | 中断控制器 - OCW₃ 的命令字举例

示例 1: 试读 8259A 内部寄存器 ISR 的值

- OCW₃ 命令字

| | | | | | | | |
|---|------|-----|---|---|---|----|-----|
| 0 | ESMM | SMM | 0 | 1 | P | RR | RIS |
| 0 | 0 | 0 | 0 | 1 | 0 | 1 | 1 |

- 设端口地址为 20H~21H

```
MOV AL, 0BH  
OUT 20H, AL ; 向控制端口发出读命令字 OCW3  
  
IN AL, 20H ; 使用输入指令读 ISR 的值
```

03 | 中断控制器-OCW₃的命令字举例

示例 2: 读 8259A 查询字, 判断当前优先级最高的中断请求

● OCW₃ 命令字

| | | | | | | | |
|---|------|-----|---|---|---|----|-----|
| 0 | ESMM | SMM | 0 | 1 | P | RR | RIS |
| 0 | 0 | 0 | 0 | 1 | 1 | 0 | 0 |

● 设端口地址为 20H~21H

```
MOV AL, 0CH  
OUT 20H, AL ; 向控制端口发出读查询字 OCW3  
IN AL, 20H ; 使用输入指令读查询字
```

● 查询字格式

| | | | | | | | |
|---|---|---|---|---|----------------|----------------|----------------|
| 1 | x | x | x | x | W ₂ | W ₁ | W ₀ |
|---|---|---|---|---|----------------|----------------|----------------|

- 若读入查询字的内容为 **10000100**, 则表明当前优先级最高的中断请求为 IR₄

03 | 中断控制器-子目录

8259A 中断控制器

- 8259A 的工作特点
- 8259A 的引脚和内部结构
- 8259A 的工作方式
- 8259A 的编程
 - 8259A 的 4 个初始化命令字
 - 8259A 的初始化编程
 - 8259A 的 3 个操作命令字
 - 8259A 命令字的使用
- 8259A 的应用
 - 8259A 的单片和级联使用
- 8259A 的应用举例

03 | 中断控制器 - 8259A命令字的使用

8259A 命令字的使用

- 8259A 共有 7 个命令字，其中 **4 个初始化命令字** ($ICW_1 \sim ICW_4$) 和 **3 个操作命令字** ($OCW_1 \sim OCW_3$)。
 - 使用 8259A 之前，通过 ICW 的设置使当前 8259A 对各类中断管理方式进行**初始化规定**
 - 在 8259A 使用过程中，通过写入 OCW，**随时**进行所需要的**设置**
- 8259A 根据各命令字的**写入端口**、**写入顺序**、以及**命令字中的标志位**实现对其区分的

03 | 中断控制器 - CPU对8259A的读写操作

CPU 对 8259A 的读写操作

- **写偶**地址端口 (20H)
 - ICW₁、OCW₂、OCW₃
- **写奇**地址端口 (21H)
 - ICW₂、ICW₃、ICW₄、OCW₁
- **读偶**地址端口 (20H)
 - IRR、ISR、查询字
- **读奇**地址端口 (21H)
 - OCW₁ (IMR)

| \overline{CS} | \overline{RD} | \overline{WR} | A ₀ | 功能 |
|-----------------|-----------------|-----------------|----------------|--|
| 0 | 1 | 0 | 0 | 写 ICW ₁ 、OCW ₂ 、OCW ₃ |
| 0 | 1 | 0 | 1 | 写 ICW ₂ 、ICW ₃ 、ICW ₄ 、OCW ₁ |
| 0 | 0 | 1 | 0 | 读 IRR、ISR、查询字 |
| 0 | 0 | 1 | 1 | 读 IMR |

03 | 中断控制器-8259A的读写命令字的区分

写入偶地址端口的命令字

- 由命令字中的标志位区分以下三个命令字

| | | | | | | | | |
|------------------|---|------|-----|---|------|----------------|----------------|-----------------|
| ICW ₁ | 0 | 0 | 0 | 1 | LTIM | 0 | SNGL | IC ₄ |
| OCW ₂ | R | SL | EOI | 0 | 0 | L ₂ | L ₁ | L ₀ |
| OCW ₃ | 0 | ESMM | SMM | 0 | 1 | P | RR | RIS |

03 | 中断控制器-8259A的读写命令字的区分

写入奇地址端口的命令字

- 由命令字中的顺序区分以下四个命令字

| | | | | | | | | |
|-------------------|----------------|----------------|----------------|----------------|----------------|-----------------|-----------------|-----------------|
| ICW ₂ | D ₇ | D ₆ | D ₅ | D ₄ | D ₃ | 0 | 0 | 0 |
| 主ICW ₃ | S ₇ | S ₆ | S ₅ | S ₄ | S ₃ | S ₂ | S ₁ | S ₀ |
| 从ICW ₃ | * | * | * | * | * | ID ₂ | ID ₁ | ID ₀ |
| ICW ₄ | 0 | 0 | 0 | SFNM | BUF | M/S# | AEOI | μPM |
| OCW ₁ | M ₇ | M ₆ | M ₅ | M ₄ | M ₃ | M ₂ | M ₁ | M ₀ |

03 | 中断控制器-子目录

8259A 中断控制器

- 8259A 的工作特点
- 8259A 的引脚和内部结构
- 8259A 的工作方式
- 8259A 的编程
 - 8259A 的 4 个初始化命令字
 - 8259A 的初始化编程
 - 8259A 的 3 个操作命令字
 - 8259A 命令字的使用
- 8259A 的应用
 - 8259A 的单片和级联使用
- 8259A 的应用举例

03 | 中断控制器-8259A的应用


单片工作方式

- 最多可以引入 **8 个中断源**的中断请求
- 主要连接为单片 8259A 与 CPU 的连接

级联工作方式

- 两级级联时，最多可使用 **9 片 8259A 接收 64 个中断源**
- 计算芯片个数和可接收中断源数目之间的关系
 - 假定芯片数目为 N ，则中断源数目 = $8N - (N - 1)$
- 主要连接为多片 8259A 之间的连接，及其与 CPU 的连接

03 | 中断控制器-单片8259A的工作方式



03 | 中断控制器-多片8259A级联的工作方式

硬件连接（非缓冲方式下）


| 引脚 | 主片 | 从片 |
|-----------------|---------------------------------------|------------------|
| IR_i | 接中断源或从片的 INT | 接中断源 |
| INT | 接 CPU 的 $INTR$ 引脚 | 接主片的某个 IR_i 引脚 |
| $INTA\#$ | 直接连于 CPU 的 $INTA\#$ 引脚，由 CPU 输出到各主从芯片 | |
| $SP\#/EN\#$ | 接高电平 | 接低电平 |
| $CAS_{2\sim 0}$ | 各从片直接接于主片上，由主片输出，从片输入 | |

03 | 中断控制器-级联缓冲比较器的作用

级联缓冲比较器的作用

- 当多片 8259A 级联时，主从片的 $CAS_2 \sim CAS_0$ 全部对应相连，主片作为输出，从片作为输入，构成 8259 A 主从式的级联工作结构
- 该级联总线用于 CPU 响应从片中断时 (第一个 INTA# 有效期间)，主片向各从片传送当前响应中断所属从片的标志码，从而选中需要提供中断类型号的从片
- 各从片把接收到的标志码与自己级联/缓冲比较器中的标志码比较；若相等，则在第二个 INTA# 有效期间，将中断类型号送到数据总线上，供 CPU 读取

03 | 中断控制器-级联缓冲比较器的作用



03 | 中断控制器-8259A级联使用时注意事项

优先级问题

- 一般**主片采用特殊全嵌套方式**，使从片的各中断源的优先级得到体现，从而实现对 8259 系统所接收的所有中断源的一般全嵌套优先级管理

中断服务结束的 EOI 命令

- 若 CPU 响应的是由从片引入的中断，则在中断服务结束时，应首先**检测当前从片是否还有其他的中断服务**
 - 若没有，则**发送两个 EOI 命令**，分别用于复位主片和从片的 ISR
 - 若有，则仅发送一个 EOI 命令，复位从片的 ISR 即可


03 | 中断控制器-8259A级联举例

示例

- 某 8086 系统中，由两片 8259A 构成级联中断系统
 - ① 主片的 IR_4 和 IR_5 有中断引入，中断类型号分别是 44H 和 45H;
 - ② 从片接主片 IR_1 ，从片的 IR_3 和 IR_6 上有中断请求，中断类型号分别是 43H 和 46H。
- 试进行初始化编程

03 | 中断控制器-8259A级联举例

硬件连接图



03 | 中断控制器-8259A级联举例

主片初始化

- 某 8086 系统中，由两片 8259A 构成级联中断系统
 - ① 主片的 IR_4 和 IR_5 有中断引入，中断类型号分别是 44H 和 45H;
 - ② 从片接主片 IR_1 ，从片的 IR_3 和 IR_6 上有中断请求，中断类型号分别是 43H 和 46H。

- 初始化所需要的命令字

| | | | | | | | | | |
|-----------|---|---|---|---|---|---|---|---|------|
| ■ ICW_1 | 0 | 0 | 0 | 1 | 0 | 0 | 0 | 1 | 11H |
| ■ ICW_2 | 0 | 1 | 0 | 0 | 0 | 0 | 0 | 0 | 40H |
| ■ ICW_3 | 0 | 0 | 0 | 0 | 0 | 0 | 1 | 0 | 02H |
| ■ ICW_4 | 0 | 0 | 0 | 1 | 0 | 0 | 0 | 1 | 11H |
| ■ OCW_1 | 1 | 1 | 0 | 0 | 1 | 1 | 0 | 1 | 0CDH |

03 | 中断控制器-8259A级联举例

主片初始化

```
MOV AL, 11H  
OUT 20H, AL ;写ICW1  
MOV AL, 40H  
OUT 21H, AL ;写ICW2  
MOV AL, 02H  
OUT 21H, AL ;写ICW3
```

```
MOV AL, 11H  
OUT 21H, AL ;写ICW4  
MOV AL, 0CDH  
OUT 21H, AL ;写OCW1
```

● 初始化所需要的命令字

| | | | | | | | | | |
|--------------------|---|---|---|---|---|---|---|---|------|
| ■ ICW ₁ | 0 | 0 | 0 | 1 | 0 | 0 | 0 | 1 | 11H |
| ■ ICW ₂ | 0 | 1 | 0 | 0 | 0 | 0 | 0 | 0 | 40H |
| ■ ICW ₃ | 0 | 0 | 0 | 0 | 0 | 0 | 1 | 0 | 02H |
| ■ ICW ₄ | 0 | 0 | 0 | 1 | 0 | 0 | 0 | 1 | 11H |
| ■ OCW ₁ | 1 | 1 | 0 | 0 | 1 | 1 | 0 | 1 | 0CDH |

03 | 中断控制器-8259A级联举例

从片初始化

- 某 8086 系统中，由两片 8259A 构成级联中断系统
 - ① 主片的 IR_4 和 IR_5 有中断引入，中断类型号分别是 44H 和 45H;
 - ② 从片接主片 IR_1 ，从片的 IR_3 和 IR_6 上有中断请求，中断类型号分别是 43H 和 46H。

● 初始化所需要的命令字

| | | | | | | | | | |
|-----------|---|---|---|---|---|---|---|---|------|
| ■ ICW_1 | 0 | 0 | 0 | 1 | 0 | 0 | 0 | 1 | 11H |
| ■ ICW_2 | 0 | 1 | 0 | 0 | 0 | 0 | 0 | 0 | 40H |
| ■ ICW_3 | 0 | 0 | 0 | 0 | 0 | 0 | 0 | 1 | 01H |
| ■ ICW_4 | 0 | 0 | 0 | 0 | 0 | 0 | 0 | 1 | 01H |
| ■ OCW_1 | 1 | 0 | 1 | 1 | 0 | 1 | 1 | 1 | 0B7H |

03 | 中断控制器-8259A级联举例

从片初始化

```
MOV AL, 11H  
OUT 0A0H, AL ;写ICW1  
MOV AL, 40H  
OUT 0A1H, AL ;写ICW2  
MOV AL, 01H  
OUT 0A1H, AL ;写ICW3
```

```
MOV AL, 01H  
OUT 0A1H, AL ;写ICW4  
MOV AL, 0B7H  
OUT 0A1H, AL ;写OCW1
```

● 初始化所需要的命令字

| | | | | | | | | | |
|--------------------|---|---|---|---|---|---|---|---|------|
| ■ ICW ₁ | 0 | 0 | 0 | 1 | 0 | 0 | 0 | 1 | 11H |
| ■ ICW ₂ | 0 | 1 | 0 | 0 | 0 | 0 | 0 | 0 | 40H |
| ■ ICW ₃ | 0 | 0 | 0 | 0 | 0 | 0 | 0 | 1 | 01H |
| ■ ICW ₄ | 0 | 0 | 0 | 0 | 0 | 0 | 0 | 1 | 01H |
| ■ OCW ₁ | 1 | 0 | 1 | 1 | 0 | 1 | 1 | 1 | 0B7H |

03 | 中断控制器-关于8259A的初始化

关于 8259A 的初始化

- 微机系统中对 8259A 芯片的**初始化是由系统软件完成**的，一般不允许用户自己设置
- 对于 8259A 的初始化操作**可以在没有配置完善操作系统的单板机上进行**，例如实验系统中
- 实际使用中，一般用户只对 8259A 的 OCW_1 和 OCW_2 进行应用，用于**中断的开放/屏蔽和发出 EOI 命令**， OCW_3 很少使用

03 | 中断控制器-子目录

8259A 中断控制器

- 8259A 的工作特点
- 8259A 的引脚和内部结构
- 8259A 的工作方式
- 8259A 的编程
- 8259A 的应用
- 8259A 的应用举例
 - 8259A 在微机系统中的应用
 - 8259A 的单片应用
 - 8259A 的级联应用
 - 8259A 与 8255A 的中断传送


03 | 中断控制器-8259A在微机系统中的应用1

单片 8259A 管理中斷

- 基本设置
 - 端口地址为 20H、21H
 - 8 个中断请求均为**边沿触发**
 - 采用**一般全嵌套方式**管理中斷源
 - 设置 8 个中斷源中斷类型号的高 5 位为 **0000 1**
- 中斷源
 - 日时钟、键盘、串口、硬盘、软盘、打印机等
- **初始化编程**需设置 ICW_1 、 ICW_2 和 ICW_4

03 | 中断控制器 - 8259A在微机系统中的应用1

单片 8259A 中断系统的硬件连接



03 | 中断控制器-8259A在微机系统中的应用1

单片 8259A 中断系统的初始化编程

;ICW₁: 边沿触发, 单片, 需要 ICW₄

MOV AL, 00010011B

| | | | | | | | |
|---|---|---|---|------|---|------|-----------------|
| 0 | 0 | 0 | 1 | LTIM | 0 | SNGL | IC ₄ |
|---|---|---|---|------|---|------|-----------------|

OUT 20H, AL

;ICW₂: 中断类型号

MOV AL, 00001000B

| | | | | | | | |
|----------------|----------------|----------------|----------------|----------------|---|---|---|
| D ₇ | D ₆ | D ₅ | D ₄ | D ₃ | 0 | 0 | 0 |
|----------------|----------------|----------------|----------------|----------------|---|---|---|

OUT 21H, AL

;ICW₄: 一般全嵌套, 非缓冲, 非自动EOI, 16位系统

MOV AL, 00000001B

| | | | | | | | |
|---|---|---|------|-----|------|------|-----|
| 0 | 0 | 0 | SFNM | BUF | M/S# | AEOI | μPM |
|---|---|---|------|-----|------|------|-----|

OUT 21H, AL


03 | 中断控制器 - 8259A在微机系统中的应用2

级联 8259A 管理中斷

- 286 以上微机中，采用**两片 8259A 级联的非缓冲方式**连接，从片的 INT 接主片的 IR_2 引脚，共容纳 **15 个中断源**
- 主片端口地址为 20H、21H，从片端口地址为 0A0H、0A1H
- 主从片的中断请求均采用**边沿触发**
- 所有中断源采用**一般全嵌套**的优先权管理方式
- 主片中断类型号的高 5 位为 **0000 1**
- 从片中断类型号的高 5 位为 **0111 0**
- 主从片的初始化均需设置 $ICW_1 \sim ICW_4$

03 | 中断控制器-8259A在微机系统中的应用2

两片 8259A 中断系统的硬件连接



03 | 中断控制器-8259A在微机系统中的应用2

主片 8259A 初始化

;ICW₁: 边沿触发, 级联, 需要 ICW₄

```
MOV AL, 00010001B
```

```
OUT 20H, AL
```

| | | | | | | | |
|---|---|---|---|------|---|------|-----------------|
| 0 | 0 | 0 | 1 | LTIM | 0 | SNGL | IC ₄ |
|---|---|---|---|------|---|------|-----------------|

;I/O端口的延时要求

```
JMP SHORT $+2
```

;ICW₂: 中断类型号

```
MOV AL, 00001000B
```

```
OUT 21H, AL
```

| | | | | | | | |
|----------------|----------------|----------------|----------------|----------------|---|---|---|
| D ₇ | D ₆ | D ₅ | D ₄ | D ₃ | 0 | 0 | 0 |
|----------------|----------------|----------------|----------------|----------------|---|---|---|

;ICW₃: 从片 INT 接主片 IR₂

```
MOV AL, 00000100B
```

```
OUT 21H, AL
```

| | | | | | | | |
|----------------|----------------|----------------|----------------|----------------|----------------|----------------|----------------|
| S ₇ | S ₆ | S ₅ | S ₄ | S ₃ | S ₂ | S ₁ | S ₀ |
|----------------|----------------|----------------|----------------|----------------|----------------|----------------|----------------|

```
JMP SHORT $+2
```

;ICW₄: 特殊全嵌套, 非缓冲, 非自动EOI, 16位系统

```
MOV AL, 00010001B
```

```
OUT 21H, AL
```

| | | | | | | | |
|---|---|---|------|-----|------|------|-----|
| 0 | 0 | 0 | SFNM | BUF | M/S# | AEOI | μPM |
|---|---|---|------|-----|------|------|-----|

03 | 中断控制器-8259A在微机系统中的应用2

从片 8259A 初始化

; ICW₁: 边沿触发, 级联, 需要 ICW₄

MOV AL, 00010001B

| | | | | | | | |
|---|---|---|---|------|---|------|-----------------|
| 0 | 0 | 0 | 1 | LTIM | 0 | SNGL | IC ₄ |
|---|---|---|---|------|---|------|-----------------|

OUT 0A0H, AL

; I/O端口的延时要求

JMP SHORT \$+2

; ICW₂: 中断类型号的初值为 70H

MOV AL, 01110000B

| | | | | | | | |
|----------------|----------------|----------------|----------------|----------------|---|---|---|
| D ₇ | D ₆ | D ₅ | D ₄ | D ₃ | 0 | 0 | 0 |
|----------------|----------------|----------------|----------------|----------------|---|---|---|

OUT 0A1H, AL

; ICW₃: 从片 INT 接主片 IR₂

MOV AL, 00000010B

| | | | | | | | |
|---|---|---|---|---|-----------------|-----------------|-----------------|
| 0 | 0 | 0 | 0 | 0 | ID ₂ | ID ₁ | ID ₀ |
|---|---|---|---|---|-----------------|-----------------|-----------------|

OUT 0A1H, AL

JMP SHORT \$+2

; ICW₄: 一般全嵌套, 非缓冲, 非自动EOI, 16位系统

MOV AL, 00000001B

| | | | | | | | |
|---|---|---|------|-----|------|------|-----|
| 0 | 0 | 0 | SFNM | BUF | M/S# | AEOI | μPM |
|---|---|---|------|-----|------|------|-----|

OUT 0A1H, AL

03 | 中断控制器-子目录

8259A 中断控制器

- 8259A 的工作特点
- 8259A 的引脚和内部结构
- 8259A 的工作方式
- 8259A 的编程
- 8259A 的应用
- 8259A 的应用举例
 - 8259A 在微机系统中的应用
 - 8259A 的单片应用
 - 8259A 的级联应用
 - 8259A 与 8255A 的中断传送

03 | 中断控制器-8259A的单片应用

CPU 每次响应 8259A 引入的某一中断时，输出显示字符串“A 8259A INTERRUPT!”，中断 10 次后退出。设该中断的中断类型号为 0AH，16 位系统，试写出控制程序

- 仅有一个中断源的中断系统，使用单片 8259A 即可
- 中断源的中断类型号为 0AH (0000 1010)
 - 8259A 的 ICW₂ 为 08H，该中断源接 IR₂ 引脚
- 程序控制
 - 每次执行**中断服务程序**时，计数值减 1
 - **主程序**在等待中断时，不断地判断计数是否满 10 次
 - 中断源响应 10 次后退出
 - ◆ 即**屏蔽 8259A 的 IR₂ 引脚**
- 处理中断需要编写**主程序和中断服务程序**

03 | 中断控制器-8259A的单片应用

主程序

- 关中断
- 初始化 8259A
- 设置中断向量
- 开中断，等待中断

中断服务程序

- 保护寄存器
- 开中断 (STI)
(或放在最后一步)
- [完成中断请求服务]
- 发 EOI 命令
- 恢复寄存器
- IRET

03 | 中断控制器-8259A的单片应用

设置中断源向量

- 使用 DOS 功能调用修改或写入 0AH 号中断向量
- 以下为 25H 号 DOS 功能调用直接写中断向量的程序段

```
MOV AH, 25H ;DOS调用的功能号送入 AH
MOV AL, 0AH ;中断类型号送入 AL
MOV AX, SEG INT_PRO
MOV DS, AX
MOV DX, OFFSET INT_PRO ;中断向量置于 DS:DX 中
INT 21H
```

03 | 中断控制器-8259A的单片应用

8259A 初始化

- 单片 8259A 使用, 16 位系统, 只有一个中断源
 - 初始化需设置: ICW_1 、 ICW_2 、 ICW_4 、 OCW_1

; ICW_1 : 边沿触发, 单片, 需要 ICW_4

MOV AL, 00010011B

| | | | | | | | |
|---|---|---|---|------|---|------|-----------------|
| 0 | 0 | 0 | 1 | LTIM | 0 | SNGL | IC ₄ |
|---|---|---|---|------|---|------|-----------------|

OUT 20H, AL

; ICW_2 : 中断类型号 0AH, 高 5 位为 08H

MOV AL, 00001000B

| | | | | | | | |
|----------------|----------------|----------------|----------------|----------------|---|---|---|
| D ₇ | D ₆ | D ₅ | D ₄ | D ₃ | 0 | 0 | 0 |
|----------------|----------------|----------------|----------------|----------------|---|---|---|

OUT 21H, AL

; ICW_4 : 设置为 16 位系统工作系统

MOV AL, 00000001B

| | | | | | | | |
|---|---|---|------|-----|------|------|-----|
| 0 | 0 | 0 | SFNM | BUF | M/S# | AEOI | μPM |
|---|---|---|------|-----|------|------|-----|

OUT 21H, AL

; OCW_1 : 仅允许 IR₂ 的中断请求

MOV AL, 11111011B

| | | | | | | | |
|----------------|----------------|----------------|----------------|----------------|----------------|----------------|----------------|
| M ₇ | M ₆ | M ₅ | M ₄ | M ₃ | M ₂ | M ₁ | M ₀ |
|----------------|----------------|----------------|----------------|----------------|----------------|----------------|----------------|

OUT 21H, AL

03 | 中断控制器-8259A的单片应用

中断次数的判断

- 等待中断的同时判断中断次数

```
LL:  NOP
     JCXZ EXIT ;判断中断次数是否为 0
     JMP LL
EXIT: MOV AL, 0FFH
     OUT 21H, AL
     HLT
```

03 | 中断控制器-8259A的单片应用


中断服务程序

数据段的定义:

```
DATA SEGMENT  
MESS DB 'A 8259A  
INTERRUPT!', 10, 13, '$'  
DATA ENDS
```

使用了寄存器: **DS**、**DX**、**AX**
CX 是主程序与中断服务程序之间的**传递参数**
需要保护的寄存器有: **DS**、**DX**、**AX**

STI



普通 EOI 命令
MOV AL, 20H
OUT 20H, AL
特殊 EOI 命令
MOV AL, 62H
OUT 20H, AL

| | | | | | | | |
|---|----|-----|---|---|----------------|----------------|----------------|
| R | SL | EOI | 0 | 0 | L ₂ | L ₁ | L ₀ |
|---|----|-----|---|---|----------------|----------------|----------------|

03 | 中断控制器-8259A的单片应用

显示字符串、记录中断次数、保护寄存器

- 调用 9 号 DOS 功能，显示字符串

```
MOV AX, DATA  
MOV DS, AX  
MOV DX, OFFSET MESS  
MOV AH, 09H  
INT 21H ;显示字符串
```

- 记录中断次数：预置计数次数 10

```
DEC CX ;计数器减 1
```

- 寄存器的保护与恢复

```
PUSH DS  
PUSH DX  
PUSH AX
```

```
POP AX  
POP DX  
POP DS
```

03 | 中断控制器-8259A的单片应用

源程序清单 1

```
DATA1 SEGMENT
    MESS DB 'A 8259A INTERRUPT!',10,13, '$'
DATA1 ENDS
CODE1 SEGMENT ASSUME CS:CODE1, DS:DATA1
INT_PRO PROC NEAR
    ;寄存器的保护
    PUSH DS
    PUSH DX
    PUSH AX
    ;使用 9 号 DOS 功能调用, 显示字符串;
    MOV AX, DATA
    MOV DS, AX
    MOV DX, OFFSET MESS
    MOV AH, 09H
    INT 21H
    ;记录中断次数
    DEC CX
    ;普通 EOI 命令
    MOV AL, 20H
    OUT 20H, AL
    ;寄存器的恢复
    POP AX
    POP DX
    POP DS
    ;开中断
    STI
    IRET
INT_PRO ENDP
```

03 | 中断控制器-8259A的单片应用

源程序清单 2

```
START: CLI
      ;设置中断向量
      MOV AX, SEG INT_PRO
      MOV DS, AX
      MOV DX, OFFSET INT_PRO
      MOV AL, 0AH
      MOV AH, 25H
      INT 21H
      ;初始化8259A
      MOV AL, 13H ;ICW1
      OUT 20H AL
      MOV AL, 08H ;ICW2
      OUT 21H, AL
      MOV AL, 01H ;ICW4
      OUT 21H, AL
      MOV AL, 0FBH ;OCW1
      OUT 21H, AL
```

```
      ;设置计数初值
      MOV CX, 10
      ;开中断
      STI
      ;等待中断的同时判断中断次数
LL:   NOP
      JCXZ EXIT
      JMP LL
      ;屏蔽 IR2
EXIT: MOV AL, 0FFH
      OUT 21H, AL
      ;返回 DOS
      MOV AX, 4C00H
      INT 21H
CODE1 ENDS
      END START
```

03 | 中断控制器-子目录

8259A 中断控制器

- 8259A 的工作特点
- 8259A 的引脚和内部结构
- 8259A 的工作方式
- 8259A 的编程
- 8259A 的应用
- 8259A 的应用举例
 - 8259A 在微机系统中的应用
 - 8259A 的单片应用
 - 8259A 的级联应用
 - 8259A 与 8255A 的中断传送

03 | 中断控制器-8259A的级联应用

8259A 的级联应用

- 设有两个可屏蔽中断源 L0、L1
- L0、L1 的中断类型号分别为 26H (00100110B)、34H (00110100B)，中断请求采用边沿触发，L1 优先级高于 L0，非缓冲工作方式，非自动 EOI 方式，8 位工作系统
 - L0 —— 用于**输入设备** (地址为 270H) 的数据输入操作，输入数据存于 **INBUF** 为首地址的 50 个字节的缓冲区中
 - L1 —— 用于**输出设备** (地址为 370H) 的数据输出操作，输出数据在 **OUTBUF** 为首地址的 50 个字节的缓冲区中
- 试设计中断管理系统，画出 8259A 的**连接图**，并编写完整的主程序及**中断服务程序**


03 | 中断控制器-8259A的级联应用

分析

- 由中断类型号 (26H 和 34H) 的**高位**可知, 中断源 L0、L1 应分别由**两片 8259A** 管理, 所以应采用**级联**工作方式
 - 管理 L0 的 8259A 的 ICW₂ 为 **20H**
 - 管理 L1 的 8259A 的 ICW₂ 为 **30H**
- 由中断类型号的**低位**又知, L0 由一个 8259A 的 **IR₆** 引入, L1 由另一个 8259A 的 **IR₄** 引入
- 而中断源 L0、L1 的**优先级顺序决定硬件连接方式**
 - 该题目的硬件连接方式并不唯一

03 | 中断控制器-8259A的级联应用


硬件连接图



- 引入 L_0 的 8259A 作为主片时，从片可由主片 $IR_0 \sim IR_5$ 中的任何一个引脚引入
- 引入 L_1 的 8259A 作为主片时，从片可由主片 $IR_5 \sim IR_7$ 中的任何一个引脚引入

03 | 中断控制器-8259A的级联应用

主程序流程图



03 | 中断控制器-8259A的级联应用

主 8259A 初始化

- 假设从片连于主片的 **IR₃** 引脚
- 需要设置 ICW₁、ICW₂、ICW₃、OCW₁
- **不需要设置 ICW₄**，因为从片只有一个中断源，不可能发生从片的中断嵌套现象

| | | | | | | | | | |
|--------------------|---|---|---|---|---|---|---|---|------|
| ■ ICW ₁ | 0 | 0 | 0 | 1 | 0 | 0 | 0 | 0 | 10H |
| ■ ICW ₂ | 0 | 0 | 1 | 0 | 0 | 0 | 0 | 0 | 20H |
| ■ ICW ₃ | 0 | 0 | 0 | 0 | 1 | 0 | 0 | 0 | 08H |
| ■ OCW ₁ | 1 | 0 | 1 | 1 | 0 | 1 | 1 | 1 | 0B7H |

L0 的中断类型为 26H (00100110B)，中断请求采用边沿触发；L1 的中断类型为 34H (00110100B)，优先级高于 L0，非缓冲工作方式，非自动 EOI 方式，8 位工作系统

03 | 中断控制器-8259A的级联应用

从 8259A 初始化

- 需要设置 ICW_1 、 ICW_2 、 ICW_3 、 OCW_1

| | | | | | | | | | |
|-----------|---|---|---|---|---|---|---|---|------|
| ■ ICW_1 | 0 | 0 | 0 | 1 | 0 | 0 | 0 | 0 | 10H |
| ■ ICW_2 | 0 | 0 | 1 | 1 | 0 | 0 | 0 | 0 | 30H |
| ■ ICW_3 | 0 | 0 | 0 | 0 | 0 | 0 | 1 | 1 | 03H |
| ■ OCW_1 | 1 | 1 | 1 | 0 | 1 | 1 | 1 | 1 | 0EFH |

L0 的中断类型为 26H (00100110B)，中断请求采用边沿触发；L1 的中断类型为 34H (00110100B)，优先级高于 L0，非缓冲工作方式，非自动 EOI 方式，8 位工作系统

03 | 中断控制器-8259A的级联应用

主程序清单 1

```
CLI
;设置 L0 中断向量
MOV AX, SEG INT0
MOV DS, AX
MOV DX, OFFSET INT0
MOV AL, 26H
MOV AH, 25H
INT 21H
;初始化主 8259A
MOV AL, 10H ;ICW1
OUT 20H AL
MOV AL, 20H ;ICW2
OUT 21H, AL
MOV AL, 08H ;ICW3
OUT 21H, AL
MOV AL, 0B7H ;OCW1
OUT 21H, AL
```

```
;设置 L1 中断向量
MOV AX, SEG INT1
MOV DS, AX
MOV DX, OFFSET INT1
MOV AL, 34H
MOV AH, 25H
INT 21H
;初始化从 8259A
MOV AL, 10H ;ICW1
OUT 0A0H AL
MOV AL, 30H ;ICW2
OUT 0A1H, AL
MOV AL, 03H ;ICW3
OUT 0A1H, AL
MOV AL, 0EFH ;OCW1
OUT 0A1H, AL
```

03 | 中断控制器-8259A的级联应用

主程序清单 2

```
;设置输入数据指针和计数器  
LEA DI, INBUF  
MOV CH, 50  
;设置输出数据指针和计数器  
LEA SI, OUTBUF  
MOV CL, 50  
;开中断  
STI  
MOV AL, 1011 0111B
```

```
;等待中断, 并判断  
LL: CMP CL, 0  
JNZ P1  
OR AL, 08H ;屏蔽IR3  
P1: CMP CH, 0  
JNZ P2  
OR AL, 40H ;屏蔽IR6  
P2: OUT 21H, AL ;OCW1  
CMP AL, 0FFH  
JZ EXIT  
JMP LL  
EXIT: HLT
```

03 | 中断控制器-8259A的级联应用

L0 的中断服务程序

- 中断服务程序流程：
 - 保护寄存器
 - 开中断
 - 输入数据
 - 发 EOI 命令
 - 恢复寄存器
 - IRET
- 先开中断：L1 的优先权高于 L0，可以发生中断嵌套现象

03 | 中断控制器-8259A的级联应用


L₀ 的中断服务程序

```
INT0 PROC
;保护寄存器
PUSH AX
PUSH DX
;开中断
STI
;输入数据并保存
MOV DX, 270H
IN AL, DX
MOV [DI], AL
```

```
;修改指针
INC DI
DEC CH
;发 EOI 命令
MOV AL, 66H ;01100110B
OUT 20H, AL
;恢复寄存器
POP DX
POP AX
;中断返回
IRET
```

```
INT0 ENDP
```

■ OCW₂



03 | 中断控制器-8259A的级联应用

L1 的中断服务程序

- 中断服务程序流程：
 - 保护寄存器
 - 输出数据
 - 发 EOI1 命令
 - 发 EOI2 命令
 - 恢复寄存器
 - 开中断
 - IRET
- 后开中断：因为该中断系统中 L1 优先级最高
- 需要发两个 EOI 命令，以复位主从片的 ISR

03 | 中断控制器-8259A的级联应用

L1 的中断服务程序

```
INT1 PROC
;保护寄存器
PUSH AX
PUSH DX
;取出数据, 并输出
MOV AL, [SI]
MOV DX, 370H
OUT DX, AL
;修改指针
INC SI
DEC CL
```

```
;发主片 EOI 命令
MOV AL, 63H ;01100011B
OUT 20H, AL
;发从片 EOI 命令
MOV AL, 64H ;01100100B
OUT 0A0H, AL
;恢复寄存器
POP DX
POP AX
;开中断
STI
;中断返回
IRET
```

```
INT1 ENDP
```

■ OCW₂

| | | | | | | | |
|---|----|-----|---|---|----------------|----------------|----------------|
| R | SL | EOI | 0 | 0 | L ₂ | L ₁ | L ₀ |
|---|----|-----|---|---|----------------|----------------|----------------|

03 | 中断控制器-子目录

8259A 中断控制器

- 8259A 的工作特点
- 8259A 的引脚和内部结构
- 8259A 的工作方式
- 8259A 的编程
- 8259A 的应用
- 8259A 的应用举例
 - 8259A 在微机系统中的应用
 - 8259A 的单片应用
 - 8259A 的级联应用
 - 8259A 与 8255A 的中断传送


03 | 中断控制器-8259A与8255A的中断传送

利用 8259A 和 8255A 设计并行打印机接口。CPU 采用中断方式将存放在缓冲区 BUF 的 256 个字符通过接口送到打印机打印

- 8255A 工作于方式 1 下，作为并行打印机的输出接口
 - 联络信号：OBF# (PC₇)、ACK# (PC₆)、INTR (PC₃)
 - 8255A 使用 PC₂ 作为打印机的自定义选通信号
- 8259A 接收 8255A 发出的中断请求，提交给 CPU 处理
 - 设中断类型为 5AH
 - 8255A 的 INTR (PC₃) 接 8259A 的 IR₂ 引脚
 - 系统中只有一个中断源，因此可定义为**自动 EOI 方式**
 - 8259A 的初始化包括：ICW₁ — ICW₂ — ICW₄ — OCW₁

03 | 中断控制器-8259A与8255A的中断传送

利用 8259A 和 8255A 设计并行打印机接口
打印机与接口之间的连接图



03 | 中断控制器-8259A与8255A的中断传送

利用 8259A 和 8255A 设计并行打印机接口 软件设计

- 主程序
 - 芯片等初始化 (8255A、8259A、CPU)
 - 设置中断向量
 - 开中断
- 中断服务程序
 - 完成数据传送、选通打印机等操作
 - **不需要发回 EOI 命令**

03 | 中断控制器-8259A与8255A的中断传送

主程序

| | | | | | | | |
|----------------|----------------|----------------|----------------|----------------|----------------|----------------|-----------------|
| 0 | 0 | 0 | 1 | LTIM | 0 | SNGL | IC ₄ |
| D ₇ | D ₆ | D ₅ | D ₄ | D ₃ | 0 | 0 | 0 |
| 0 | 0 | 0 | SFNM | BUF | M/S# | AEOI | μPM |
| M ₇ | M ₆ | M ₅ | M ₄ | M ₃ | M ₂ | M ₁ | M ₀ |

;关中断

CLI

;8255A 初始化,A口方式1输出

MOV AL, 10100000B

OUT 63H, AL

;禁止打印机工作,C口置复位

MOV AL, 00000101B

OUT 63H, AL

;8259A 初始化

MOV AL, 00010011B ;ICW₁

OUT 20H, AL

MOV AL, 01011000H ;ICW₂

OUT 21H, AL

MOV AL, 00000011B ;ICW₄

OUT 21H, AL

MOV AL, 11111011B ;OCW₁

OUT 21H, AL

;数据指针初始化

LEA SI, BUF

MOV CX, 256

;设置中断向量

MOV AH, 25H

MOV AL, 5AH

MOV DX, SEG SHUCHU

MOV DS, DX

MOV DX, OFFSET SHUCHU

INT 21H

;开中断,等待中断

STI

MOV AL, 00000111B

OUT 63H, AL

LL: STI

CMP CX, 0

JNZ LL

03 | 中断控制器-8259A与8255A的中断传送

中断服务程序

```
SHUCHU PROC
    PUSH AX          ;保护通用寄存器
    MOV AL, [SI]    ;输出打印数据
    OUT 60H, AL
    INC SI          ;修改数据指针和计数值
    DEC CX
    MOV AL, 00000100B
    OUT 63H, AL    ;发选通信号
    NOP
    NOP
    INC AL
    OUT 63H, AL    ;禁止选通打印机
    POP AX         ;恢复通用寄存器
    IRET
SHUCHU ENDP
```

若需要 EOI 命令, 则加上
MOV AL, 20H
OUT 20H, AL

本章小结

- 理解有关中断的**基本概念**
 - 中断向量、中断类型号、中断优先权、中断嵌套
- 熟悉 INTR 的中断处理过程
 - 特别是 INTR 的中断响应周期
- 掌握 8259A 面向 CPU 和中断源，有关中断的外部引脚
- 理解 8259A 中的**三个寄存器**的作用和设置
- 掌握 8259A 的 **4 个初始化命令字**和 **2 个操作命令字**，并能够灵活使用，进行初始化程序和简单处理程序的编写
- 理解 8259A 的**单片**和**级联**方式的连接和使用要点



河南大學
Henan University



Q&A

主讲教师：舒高峰

电子邮箱：gaofeng.shu@henu.edu.cn

联系电话：13161693313